



产品名称 MK7XXXX系列芯片

标题

如何使用MK7XXXX系列芯片的TIME外部RTCC计数功能

简介

有时在写程序的时候，需要用到TIME外部RTCC计数功能，这里以MK7A11P为例，介绍MKT芯片的TIME外部RTCC计数功能的使用方法。

对MK7A11P来说，与TIME外部RTCC计数功能有关的寄存器有SELECT选择寄存器，其具体定义如下：

Name	Bit7	Bit6	Bit5	Bit4	Bit3	Bit2	Bit1	Bit0
SELECT	TMR0_EN	AUTO	SUR0	EDGE0	PSA	PS2	PS1	PS0

Bit	Symbol	Description				
2~0	PS2~PS0	PS2	PS1	PS0	TMR0 rate	WDT rate
		0	0	0	1:2	1:1
		0	0	1	1:4	1:2
		0	1	0	1:8	1:4
		0	1	1	1:16	1:8
		1	0	0	1:32	1:16
		1	0	1	1:64	1:32
		1	1	0	1:128	1:64
		1	1	1	1:256	1:128
3	PSA	PSA: Prescaler assignment bit 1: Prescaler assigned to WDT 0: Prescaler assigned to TMR0				
4	EDGE0	EDGE0: TMR0 source signal edge control bit 1: increment when H→L transition on external clock 0: increment when L→H transition on external clock				
5	SUR0	SUR0: TMR0 clock source bit 1: External clock input 0: (internal clock) / 4				
6	AUTO	AUTO: Auto pre-load TMR0 data 1: enable 0: disable				
7	TMR0_EN	TMR0_EN: TMR0 enable/disable 0: disable 1: enable				

应该注意，SELECT寄存器是只写寄存器，其控制位的设定由select指令完成，例如：

```
Movla b'11100010'
```

```
Select
```

执行上述指令后，SELECT寄存器里面的内容将为b'11100010'，预分频分配给TMR0，预分频比为1:8，TMR0采用外部时钟（来自RTCC管脚）计数，上升沿有效。请注意MK7A11P的TMR0是递减的。

由于RTCC管脚与PB2共用，所以要进行相关的设置，与此有关的寄存器是PB_PDM，其结构图如下，设定方法只要将对应的RTCCE置1即可。

Register	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0
PB_PDM	RTCCE	INTE	--	--	--	DB2	DB1	DB0

线路图见图1，基本功能要求如下：

由PB1提供周期为250us的方波作为TMR0的外部时钟

由TMR0来控制LED1的开关切换（也就是亮1s，灭1s这样循环切换）




图1 MK7A11P的IO唤醒试验线路图

DEMO程序

汇编程序文档



```
;-----  
#include "mk7a11p_hw.inc" ;编译该文档需包含"mk7a11p_hw.inc"文件  
;-----  
;芯片型号 (mk7a11p)  
;-----  
;配置寄存器设置说明 (CONFIG)  
;1-----FOSC=RC ;LS,NS,HS,RC  
;2-----INRC=ON ;ON,OFF  
;3-----CPT=OFF ;ON,OFF  
;4-----WDTE=Disable ;Enable,Disable  
;5-----LV=Low Vol Reset ON ;Low Vol Reset ON,Low Vol Reset OFF  
;6-----RESET=...input... ;...input.....reset...  
;-----  
flag equ 0x23 ;标志  
del_r equ 0x24 ;延时  
;-----  
#define flag_key flag,0 ;按键未松开标志  
#define flag_t flag,1 ;重新计时标志  
;-----  
org 0x3ff ;mk7a11p的复位向量地址定义  
lgoto main ;跳转到主程序入口  
;-----  
org 0x100 ;主程序入口地址定义  
main  
movla 0x20  
movam fsr ;将0x20送入fsr寄存器  
;-----  
clear_ram  
;利用indf和fsr来进行间接寻址  
;对0x20-0x2f的RAM进行clear  
clr indf  
mov fsr,a  
andla b'00111111';将无关的数据虑除  
xorla 0x2f  
btsc status,z  
lgoto $+3  
inc fsr,m
```



```
lgoto      clear_ram

clr  fsr ;使用fsr时要注意bank的归位

;-----

;PortA端口方向及状态设定

movla      b'11111101'

iodir      porta

clr  porta

clr  pa_pdm

;-----

;PortB端口方向及状态设定

movla      b'11111001'

iodir      portb

clr  portb

clr  pb_pod

clr  pb_pup

;-----

;将PB2/RTCC作为TMR0外部时钟的输入脚

movla      b'10000000'

movam      pb_pdm

;-----

;PortB端口8个IO唤醒禁止

movla  b'00000000'

movam  wake_up

;-----

;配置TMR0,预分频比为1:16

;TMR0初始值为250

;时钟源为外部时钟

movla      b'11100011'

select

movla  .250

movam  tmr0

;-----

;禁止中断,清除中断标志

clr  irqm

clr  irqf

;-----

loop    ;程序循环入口
```



```
    btss irqf,0
    lgoto jump_0

    bc irqf,0
    movla b'00000010'
    xor porta ;时间到进行LED开关切换
    lgoto jump_1
;-----
jump_0
;保持时间等长

    nop
    nop
    nop
    nop
;-----
jump_1
    bs portb,1
;-----
;high--125us
    inc del_r,m
    movla .20
    xor del_r,a
    btss status,z
    lgoto $-4
    clr del_r
    nop
    nop
    nop
    nop
    bc portb,1
;-----
;low--125us
    inc del_r,m
    movla .19
    xor del_r,a
    btss status,z
    lgoto $-4
    clr del_r
```



```
    nop
    lgoto    loop
;-----
    end
mk7a10p_hw.inc文档
;-----Define special register(Define SFR) -----
indf equ    0x00
tmr0  equ    0x01
pc    equ    0x02
status equ    0x03
fsr   equ    0x04
porta equ    0x05    ;porta(0-3)
portb equ    0x06    ;portb(0-7)
;-----
irqm  equ    0x09
irqf  equ    0x0a
;-----
pa_pdm equ    0x0b
pb_pup equ    0x0c
pb_pdm equ    0x0d
pb_pod equ    0x0e
wake_up equ    0x0f
;-----Define [status Register] special bit-----
c      equ    0
dc     equ    1
z      equ    2
pd     equ    3
to     equ    4
;-----Define [irqm Register] special bit-----
tm0m equ    0
extm equ    1
intm equ    7
;-----Define [irqf Register] special bit-----
tm0f equ    0
extf equ    1
;-----Define [pa_pdm Register] special bit-----
da0 equ    0
```



```
da1 equ 1
da2 equ 2
da3 equ 3
;-----Define [pb_pup Register] special bit-----
ub0 equ 0
ub1 equ 1
ub2 equ 2

ub4 equ 4
ub5 equ 5
ub6 equ 6
ub7 equ 7
;-----Define [pb_pdm Register] special bit-----
db0 equ 0
db1 equ 1
db2 equ 2

inte equ 6
rtce equ 7
;-----Define [pb_pod Register] special bit-----
ob0 equ 0
ob1 equ 1
ob2 equ 2

ob4 equ 4
ob5 equ 5
ob6 equ 6
ob7 equ 7
;-----Define [wake_up Register] special bit-----
en0 equ 0
en1 equ 1
en2 equ 2
en3 equ 3
en4 equ 4
en5 equ 5
en6 equ 6
en7 equ 7
```



芯睿科技股份有限公司
mikkon technology limited

MK7A10P
8 bit microcontroller

;